Drexel
University

An Interactive BASIC Program to Calculate Shallow Water, Limited Fetch Wave Conditions

J. Richard Weggel
Scott L. Douglass

Hydraulics & Hydrology Laboratory
Report 85-1
September 1985

NJSG-85-143

An Interactive BASIC Program to Calculate Shallow Water, Limited Fetch Wave Conditions

by
J. Richard Weggel
Scott L. Douglass

Report Number 85-1
Hydraulics and Hydrology Laboratory
Civil Engineering Department
Drexel University
Philadelphia, Pennsylvania 19104

September 1985

PREFACE

This is the first of what is hoped to be a continuing series of reports emanating from the Drexel University Department of Civil Engineering's Hydraulics and Hydrology Laboratory.

This report describes an interactive computer program written in the BASIC language which was developed during the course of the data analysis for a project on the performance of a perched beach along the shores of the Delaware Bay at Slaughter Beach, Delaware. The project was funded by NOAA's Office of Sea Grant through the New Jersey Marine Sciences Consortium with cost sharing by Drexel University. Sea Grant publication number is NA83AA-D-00034. Since the program was developed partly with public funds, it is in the public domain and is here made available for general use. Comments on this report and the computer program are solicited. Program users are encouraged to submit comments and suggestions for its improvement.

CONTENTS

An Interactive BASIC Program to Calculate Shallow Water, Limited Fetch Wave Conditions

by
J. Richard Weggel and Scott L. Douglass

INTRODUCTION

The U.S. Army Corps of Engineers Shore Protection Manual (1984) presents procedures for estimating the height and period of wind-generated waves in shallow water. These procedures can be used for forecasting waves which could occur in the future due to assumed design windspeeds and durations or for hindcasting waves which have occured in the past as the result of given historical wind data.

This report presents an interactive BASIC computer program for calculating shallow water, limited fetch wave conditions. The program uses the procedures outlined in the Shore Protection Manual. The program was originally written for a TRS-80 Model III computer but is also completely compatible with the Microsoft BASIC used by IBM-compatible computers. This report describes the required input and output in detail and also describes the equations used in the program. The program listing is in Appendix A and an example program output is in Appendix B.

INPUT

English units of measure are used for the interactive input, however, with appropriate changes the program could be modified for metric units. Information requested by the interactive program are:

1. Fetch length (in feet) - Fetch is the distance the wind blows across the water.

2. Water depth (in feet) - The average water depth along the fetch.

3. Minimum, maximum, and incremental wind speeds of interest (in mph) - This is related to the desired output described in the next section (see Appendix B). Wave heights and periods are presented for wind speeds starting at the minimum wind speed entered up to the maximum wind speed in steps of the incremental windspeed.

4. Height above ground or water level at which the wind speed was measured (in feet).

5. Time-period over which the observed wind speed was averaged (in minutes).

6. Whether the wind was measured over the land or the water.

7. Air temperature (in degrees Celsius).

8. Water temperature (in degrees Celsius).

OUTPUT

An example of the program output is shown in Appendix B. The first seven lines are echo prints of the input parameters described above. The range of the wind speeds for which wave conditions are desired are given in the left-hand column. The range is entered as input (#3 above). Windspeed from the minimum to the maximum in steps of the windspeed increment are printed out. Output table column headings are defined in the footnotes to the output (see Appendix B). Printing of the definition footnotes is optional.

EQUATIONS

A brief discussion of the equations follows. The Shore Protection Manual and Resio and Vincent (1977) provide a more detailed description of the method.

WAVE GENERATION EQUATIONS

The Shore Protection Manual's shallow-water wave generation equations are:

$$H' = 0.283 \tanh (0.530\, d'^{3/4}) * \tanh \{ (0.00565\, F'^{1/2}) / \tanh (0.530\, d'^{3/4}) \} \quad (1)$$

$$T' = 7.54 \tanh (0.833\, d'^{3/8}) * \tanh \{ (0.0379\, F'^{1/3}) / \tanh (0.833\, d'^{3/8}) \} \quad (2)$$

$$t' = 5.37\, T'^{7/3} \quad (3)$$

in which the dimensionless variables; d', F', H', T', and t'; are defined as follows:

d' = dimensionless depth = $(gd) / (U_a^2)$

F' = dimensionless fetch = $(gF) / (U_a^2)$

H' = dimensionless wave height = $(gH) / (U_a^2)$

T' = dimensionless wave period = $(gT) / U_a$

t' = dimensionless duration = $(gt) / U_a$

where:
- g - acceleration due to gravity
- d - water depth
- U_a - adjusted windspeed
- F - fetch length
- H - significant wave height
- T - wave period
- t - duration

EQUATION FOR H_{10}

The wave height result from Equation 1 is assumed to represent the significant wave height, H_s, defined as the average height of the highest one-third of the waves. The significant wave height can be related to other waves in the wave spectrum under the assumption that the heights are distributed according to a Rayleigh Distribution. The program presents the average of the highest 10% of the waves, H_{10}, using Equation 4 which is a result of the assumption of a Rayleigh distribution of heights.

$$H_{10} = 1.28\ H_s \tag{4}$$

WIND CORRECTIONS

Several corrections to the measured windspeed are required to make it consistent with that used in the wave generation equations. The windspeed that should be used in the wave generation equations is called the adjusted windspeed, U_a. The corrections depend on where the windspeed was measured, over how long a period the recorded windspeed was averaged, on the fetch length, and on the effectiveness of the energy exchange between the wind and water which is related to the air-sea temperature difference. Several of these corrections relate to the atmospheric boundary layer, the approximately 1000 meter thick layer above the earth's surface in which the boundary shear distorts the geostrophic wind field. See Resio & Vincent (1977) for a more detailed explanation of the source of these corrections.

Height of Observation

Observed windspeed depends upon the height at which it is measured. An idealization of this vertical variation in windspeed is shown in Figure 1. A common base height for wind data is 10 meters above the ground. Equation 5 can be used to correct windspeeds measured at other heights less than 20 meters above the ground.

$$U(10) = U(z)\ (10/z)^{1/7} \tag{5}$$

where $U(10)$ - windspeed 10 meters above ground
$U(z)$ - windspeed z meters above ground
z - height above ground atwhich wind measurement was made.



Figure 1. Vertical variation in windspeed (from SPM).

Duration

Windspeed is not steady and thus the windspeed reported is usually an average taken over some time period. Simiu & Scanlan (1978) empirically related windspeeds obtained for different averaging intervals. Equations 6a and 6b fit their data, are shown in Figure 2, and are used to convert a windspeed to an equivalent windspeed of a different averaging time period.

$$U_t / U_{3600} = 1.277 + 0.296 \tanh\{0.9 \log(45/t)\} \quad (1 \text{ sec} < t < 3600 \text{ sec}) \qquad (6a)$$

$$U_t / U_{3600} = -0.15 \log(t) + 1.5334 \quad (3600 \text{ sec} < t < 36000 \text{ sec}) \qquad (6b)$$

where: t - duration of interest (seconds)

U_t - equivalent windspeed of duration t

U_{3600} - one-hour average windspeed

Equations 6a and 6b can be used to convert windspeeds averaged over any time period to any other averaging time by first converting the observation to a one-hour averaging time.

Overland Measurement

The surface roughness felt by wind due to the earth's surface is a function of surface type, terrain, and vegetation. Wind "feels" more surface friction when passing over land than water. Ideally, wind for wave forecasting should be measured over the water. However, such data are often not available and wind data from a nearby land site, usually an airport, are used. Figure 3, from Vincent & Resio (1977) can be used to correct the windspeed for overland measurements. Equation 7 is an approximating function to the curve in Figure 3.

$$U_w = 2.4 * U_l * [U_l/1.689]^{(-.2737)} \qquad (7)$$

where U_w = windspeed over water (ft/s)

U_l = windspeed over land (ft/s)



Figure 2. Ratio of windspeed of any duration, U_t, to the 1-hour windspeed, U_{3600}. (from SPM)



(after Resio & Vincent, 1977)

Figure 3. Correction for overland wind measurement (from SPM).

A lower limit for the correction of $U_w = 0.9 * U_1$ is used. When the fetch length is less than 10 miles, a correction of $U_w = 1.2 * U_1$ is used.

Non-Constant Coefficient of Drag

Wave generation is a function of the drag or stress of the wind on the water. Wind stress is not linearly related to wind speed. Windspeed is adjusted for this non-constant coefficient of drag by

$$U_a = 0.864 * (U_w/1.4667)^{1.23} \qquad (8)$$

where U_a - windspeed adjusted for non-constant coefficient of drag (ft/sec)
U_w - windspeed over water (ft/sec)

Air-Sea Temperature Difference

The difference in temperature between the water and the air relates to the stability of the boundary layer which influences the effectiveness of the wind in generating waves. A windspeed correction to account for this effect is given in Figure 4. An approximation to the curve in Figure 4 is given by Equation 9. If temperature information is not available, a constant correction of $U' = 1.1 * U$ is used.

$$U' = \{1 + 0.06878 * abs(T_w - T_a)\}^{0.3881} * sign(T_w - T_a) * U \qquad (9)$$

where U' = windspeed adjusted for air-sea temperature difference (in ft/sec)
T_w = water temperature (in degrees Celsius)

T_a = air temperature (in degrees Celsius)

U = winspeed not adjusted for air-sea temperature difference (in ft/sec)
abs = absolute value

$sign(T_w - T_a)$ = +1 when $(T_w - T_a) > 0$
= -1 when $(T_w - T_a) < 0$
= +1 when $(T_w - T_a) = 0$



Figure 4. Correction for air-sea temperature difference (from SPM).

ITERATION PROCEDURE

The wave generation equations, Equations 1, 2, and 3, are for fetch-limited, fully arisen sea conditions. The program assumes that the averaging period used to determine the windspeed is the same as the minimum duration, the time it takes the wave field to reach fully arisen conditions (given by Equation 3). With an iterative procedure, Equation 6 and the wave generation equations can be used to determine duration-limited conditions.

The program calculates the fully-developed wave conditions due to the one-hour averaged windspeed and then calculates the minimum duration which would be required to reach those conditions for that windspeed. The windspeed is adjusted using Equation 6 to an equivalent windspeed averaged over that minimum duration. This new windspeed is inserted back into the wave generation procedure to determine wave conditions. This iteration continues until the duration of the equivalent averaged windspeed equals the minimum required duration. Only several iterations are required for convergence.

EXAMPLE PROBLEM

Given: An anemometer measures a sustained 50 mph wind. The anemometer is 25 feet above land and the windspeed observation is a ten-minute average windspeed. In the direction of the wind, the fetch is 75000 feet and the average depth across that fetch is 30 feet. Air temperature is 4.4 degrees Celcius and water temperature is 18.3 degrees Celcius.

Find: Significant wave height and period.

Solution: From the example output listed in Appendix B,

$$H_s = 6.4 \text{ feet}$$
$$T_s = 4.8 \text{ seconds}$$

ACKNOWLEDGEMENTS

This report and the computer program it describes were prepared as part of the data analysis on a project entitled, "Evaluation of the Performance of a Perched Beach" funded by NOAA's Office of Sea Grant through the New Jersey Marine Sciences Consortium with cost sharing by Drexel University. Their support is gratefully acknowledged.

REFERENCES

Resio, D.T., and Vincent, C.L., "Estimation of Winds Over the Great Lakes," Journal of the Waterway, Port, Coastal and Ocean Engineering Division, Proceedings of the American Society of Civil Engineers, vol. 103, No. WW2, May 1977.

Shore Protection Manual, U.S. Army Engineer Waterways Experiment Station Coastal Engineering Research Center, Vicksburg, Mississippi, Fourth edition, 1984.

Simiu, E., and Scanlan, R.N., Wind Effect on Structures: An Introduction to Wind Engineering, New York, Wiley, p. 62, 1978.

```
10 'Program name  =  WAVEGEN
20 '
30 'Programmed by:  J. Richard Weggel, Ph.D., P.E.
40 '                Department of Civil Engineering
50 '                Drexel University
60 '                Philadelphia, PA  19104
70 '
80 'An interactive program to calculate wave conditions in shallow water
90 'limited fetch conditions.
100 'This program is described in detail in "An Interactive BASIC Program to
     Calculate Shallow Water, Limited Fetch Wave Conditions," Report No. 85-1,
110 'Hydraulics and Hydrology Laboratory, Civil Engineering Department,
     Drexel University, by J. Richard Weggel and Scott L. Douglass.
120 'Programmed in BASIC which is compatible with Radio Shack's TRS-80 Model III
     and any IBM-compatible with Microsoft's BASIC.
130 '
140 A$="####.##  "
150 CLS
160 DEF FNTH(A)=(EXP(A)-EXP(-A))/(EXP(A)+EXP(-A))
170 LPRINT "WAVE FORECASTING EQUATIONS"
180 '************************************************************
190 '*                    INPUT INFORMATION                     *
200 '************************************************************
210 INPUT"Enter the fetch length in feet.";F
220 LPRINT:LPRINT USING"Fetch length = #######  feet.";F
230 INPUT"Enter the average water depth in feet.";D
240 LPRINT:LPRINT USING"Average water depth = ####.# feet.";D
250 INPUT"Enter the minimum, maximum and incremental wind speeds in mph.";UM,UX,
UI
260 INPUT"Enter the elevation (in feet) at which the wind speed was measured";Z
270 LPRINT:LPRINT USING"The elevation at which the wind speed was measured is ##
#.## feet";Z
280 INPUT"Enter the time period (in minutes) over which the wind speed was avera
ged.";TO
290 LPRINT:LPRINT USING"The time period over which the wind speed was averaged i
s ####.## minutes.";TO
300 INPUT"Was the wind measured over land or over water? L or W";C$
310 LPRINT
320 IF C$="l" OR C$="L" THEN LPRINT"Wind was measured over land." ELSE LPRINT"Wi
nd was measured over water."
330 INPUT"What is the air temperature in degrees Celsius. If unknown, enter 999.
";TA
340 LPRINT:LPRINT USING"The air temperature is ###.# degrees Celsius.";TA
350 INPUT"What is the water temperature in degrees Celsius. If unknown, enter 99
9";TW
360 LPRINT:LPRINT USING"The sea temperature is ###.# degrees Celsius.";TW
370 '************************************************************
380 '*                    PRINT TABLE HEADINGS                  *
390 '************************************************************
400 LPRINT:LPRINT
410 LPRINT "   U       U10      U10      Ua       Td       Ts       H        H1
0"
420 LPRINT "  unadj.   unadj.  adjust.  adjust."
430 LPRINT "  (mph)    (mph)    (mph)    (mph)    (min)    (sec)    (ft)     (ft
)"
440 LPRINT"------------------------------------------------------------------------
--"
```

```
450 '************************************************************
460 '*                  BEGIN CALCULATIONS                      *
470 '************************************************************
480 TO=TO*60                          'Convert averaging time to sec
490 FOR I=UM TO UX STEP UI            'Loop with wind velocities
500 U=I
510 U=U*1.4667                        'Change to ft/sec.
520 U10=(32.81/Z)^.143*U              'Correct for elevation.
530 UZ=U10                            'Retain original U at 10 m.
540 '************************************************************
550 '*               CORRECT WIND FOR DURATION                  *
560 '************************************************************
570 IF TO>3600 THEN 600
580 U3600=U10/(1.277+.296*FNTH(.3909*LOG(45/TO)))
590 GOTO 640
600 U3600=U10/(-.06514*LOG(TO)+1.5334)
610 '*********************************************************
620 '*          CORRECT WIND FOR OVERLAND MEASUREMENT        *
630 '*********************************************************
640 IF C$="W" OR C$="w" THEN GOTO 680
650 IF F<52800! THEN UW=1.1*U10 ELSE UW=2.4*U10*(U10/1.689)^(-.2737)
660 IF UW<.9*U10 THEN UW=.9*U10
670 GOTO 720
680 UW=U10
690 '*********************************************************
700 '*   CORRECT WIND FOR NON-CONSTANT COEFFICIENT OF DRAG   *
710 '*********************************************************
720 UPA=.864*(UW/1.466667)^1.23
730 '*********************************************************
740 '*    CORRECT WIND FOR AIR-SEA TEMPERATURE DIFFERENCE    *
750 '*********************************************************
760 IF F>52800! THEN 780 ELSE UA=UPA
770 GOTO 820
780 IF TA=999 OR TW=999 THEN UA=1.1*UPA ELSE UA=(1+.06878*(ABS(TW-TA))^.3881*SGN
(TW-TA))*UPA
790 '*********************************************************
800 '*              WAVE GENERATION EQUATIONS                *
810 '*********************************************************
820 GFU2=32.17*F/UA/UA               'dimensionless fetch
830 GDU2=32.17*D/UA/UA               'dimensionless depth
840 N=.0379*GFU2^.33
850 B=FNTH(.833*GDU2^.375)
860 TS=UA*.2344*B*FNTH(N/B)                'significant period in sec
870 TD=.2782*UA*(32.17*TS/UA)^(7/3)*60  'minimum duration in sec
880 IF TD>3600 THEN 910
890 UTD=U3600*(1.277+.296*FNTH(.3909*LOG(45/TD)))  'Correct velocity for
calculated duration.
900 GOTO 920
910 UTD=U3600*(-.06514*LOG(TD)+1.5334)
920 UN=UTD
930 ER=ABS((UN-U10)/UN)              'Calculate fractional error
940 IF ER>.001 THEN 950 ELSE 970
950 U10=UN
```

```
960 GOTO 640
970 N2=.00565*GFU2^.5                   'Calculate wave height info.
980 B2=FNTH(.53*GDU2^.75)
990 HS=UA*UA*.0088*B2*FNTH(N2/B2)  'significant wave height
1000 H10=HS*1.28                      'ave. height of max. 10% of waves
1010 '*********************************************************
1020 '*     PRINT OUT WIND SPEEDS AND WAVE PARAMETERS        *
1030 '*********************************************************
1040 LPRINT USING A$;U/1.46667,UZ/1.466667,U10/1.466667,UA/1.466667,TD/60,TS,HS,
H10
1050 NEXT I
1060 LPRINT"-----------------------------------------------------------------
---"
1070 LPRINT:LPRINT
1080 LPRINT:INPUT"Do you want to process more data? Y or N";B$
1090 IF B$="y" OR B$="Y" THEN 150
1100 IF B$="N" OR B$="n" THEN 1110 ELSE 1080
1110 INPUT"Do you want to print out footnote definitions? Y or N";C$
1120 IF C$="Y" OR C$="y" THEN 1140
1130 IF C$="N" OR C$="n" THEN 1240 ELSE 1110
1140 LPRINT"  U = unadjusted windspeed in mph."
1150 LPRINT"  U10 (unadj.) = windspeed corrected to 10 meter elevation in mph."
1160 LPRINT"  U10 (adjust.) = 10 meter windspeed adjusted so that averaging time
 equals"
1170 LPRINT"       minimum duration."
1180 LPRINT"  Ua = adjusted windspeed factor used in wave forecasting equations.
"
1190 LPRINT"  Td = minimum duration in minutes (time wind must blow to generate"

1200 LPRINT"       fully arisen conditions)."
1210 LPRINT"  Ts = significant wave period in seconds."
1220 LPRINT"  Hs = significant wave height in feet."
1230 LPRINT"  H10 = average height of the highest 10% of waves in feet."
1240 PRINT:PRINT"Thank-you. Goodbye."
1250 END
```

WAVE FORECASTING EQUATIONS

Fetch length = 75000 feet.

Average water depth = 30.0 feet.

The elevation at which the wind speed was measured is 25.00 feet

The time period over which the wind speed was averaged is 10.00 minutes.

Wind was measured over land.

The air temperature is 4.4 degrees Celsius.

The sea temperature is 18.3 degrees Celsius.

U unadj. (mph)	U10 unadj. (mph)	U10 adjust. (mph)	Ua adjust. (mph)	Td (min)	Ts (sec)	H (ft)	H10 (ft)
5.00	5.20	4.62	8.47	169.32	2.05	0.82	1.05
10.00	10.40	9.32	15.86	148.04	2.77	1.63	2.09
15.00	15.59	14.12	23.00	127.62	3.21	2.34	2.99
20.00	20.79	18.99	29.96	112.77	3.54	2.98	3.81
25.00	25.99	23.90	36.80	101.72	3.81	3.58	4.58
30.00	31.19	28.85	43.54	93.16	4.04	4.14	5.30
35.00	36.39	33.84	50.20	86.32	4.24	4.67	5.98
40.00	41.59	38.84	56.78	80.70	4.42	5.17	6.62
45.00	46.78	43.91	64.59	75.15	4.61	5.74	7.34
50.00	51.98	49.03	73.98	69.64	4.83	6.39	8.17
55.00	57.18	54.18	83.65	64.95	5.02	7.02	8.99
60.00	62.38	59.40	93.67	60.88	5.21	7.64	9.78
65.00	67.58	64.41	103.47	57.50	5.38	8.22	10.52
70.00	72.78	69.42	113.46	54.51	5.55	8.79	11.25
75.00	77.97	74.43	123.62	51.86	5.70	9.33	11.95

U = unadjusted windspeed in mph.
U10 (unadj.) = windspeed corrected to 10 meter elevation in mph.
U10 (adjust.) = 10 meter windspeed adjusted so that averaging time equals minimum duration.
Ua = adjusted windspeed factor used in wave forecasting equations.
Td = minimum duration in minutes (time wind must blow to generate fully arisen conditions).
Ts = significant wave period in seconds.
Hs = significant wave height in feet.
H10 = average height of the highest 10% of waves in feet.